

Corrigé du TD Maple 2

1 Exercice

Définition des matrices A et B (après avoir chargé la librairie d'algèbre linéaire : `with(linalg)`) :

```
> A:=matrix(4,4,[1,0,0,0,0,-1,0,0,0,0,0,1,0,0,0,0]):
  B:=matrix(4,4,[-1,0,0,0,-1,1,1,-1,-1,0,0,0,-1,0,0,0]):
```

1. On crée une matrice générique P , puis la matrice $AP - PB$. Il s'agit de résoudre le système d'équations qui dit que tous les coefficients de cette matrice sont nuls. On commence donc par créer l'ensemble des coefficients de cette matrice. On résout ensuite cet ensemble d'équations. Ce qu'on a obtenu est en fait un ensemble de *nombres*, et non d'*équations*. Mais, lorsque le signe « = » ne figure pas dans ce qui doit être une équation, MAPLE ajoute automatiquement « = 0 ». Attention à ne pas oublier de forcer l'évaluation des matrices avec la commande `evalm`!

```
> P:=matrix(4,4):
  APPB := evalm(A&*P - P&*B):
  ens := convert(APPB,set):
  sol := solve(ens):
  Psol := subs(sol,evalm(P));
```

$$Psol := \begin{bmatrix} P_{1,1} & -2P_{1,1} & -2P_{1,1} & 2P_{1,1} \\ P_{2,1} & 0 & 0 & 0 \\ P_{3,1} & 0 & P_{3,3} & -P_{3,1} - P_{3,3} \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

On a ainsi toutes les matrices solution. On constate en particulier que l'ensemble des solutions forme un espace vectoriel (c'était évident avant tout calcul) de dimension 4.

2. Il s'agit de trouver, parmi les matrices solutions, lesquelles sont inversibles :

```
> det(Psol);
```

0

(ce qui était évident à cause de la dernière ligne, formée uniquement de 0). Donc aucune matrice inversible n'est solution.

3. a)

```
> mu := minpoly(B,X);
```

$$\mu := -X + X^3$$

b) Autant MAPLE sait calculer un le reste pour n'importe quel entier *effectif* n , aussi grand soit-il, autant il ne sait pas faire lorsque le degré est un paramètre (essayez donc de calculer le reste dans la division de X^n par μ en posant la division...):

```
> quo(X^{17},Mu,X); quo(X^n,Mu,X);
```

$$X^{14} + X^{12} + X^{10} + X^8 + X^6 + X^4 + X^2 + 1$$

Error, (in quo) arguments must be polynomial in X

Il faut donc se débrouiller autrement. L'égalité $X^n = Q_n\mu + R_n$ (en notant Q_n le quotient) montre que, pour toute racine r de μ , on a $R_n(r) = r^n$. Comme R_n est de degré inférieur ou égal à 2 et que le polynôme μ a trois racines distinctes, R_n peut être vu comme un polynôme interpolateur. Ces racines sont évidentes, mais on peut même demander à MAPLE de les calculer :

```
> racines:=[solve(mu)];
   ord:=map(t->t^n,racines);
   R:=interp(racines,ord,X);
```

$$racines := [0, 1, -1]$$

$$ord := [0, 1, (-1)^n]$$

$$Rn := \left(\frac{1}{2}(-1)^n + \frac{1}{2}\right) X^2 + \left(-\frac{1}{2}(-1)^n + \frac{1}{2}\right) X$$

c) Le polynôme μ étant annulateur de B , on a $\mu(B) = 0$. La substitution $X \leftarrow B$ dans l'égalité $X^n = Q_n\mu + R_n$ donne alors $B^n = 0 + R_n(B)$.

```
> Bn:=evalm(subs(X=B,Rn));
```

$$Bn := \begin{bmatrix} (-1)^n & 0 & 0 & 0 \\ \frac{1}{2}(-1)^n - \frac{1}{2} & 1 & 1 & -1 \\ (-1)^n & 0 & 0 & 0 \\ (-1)^n & 0 & 0 & 0 \end{bmatrix}$$

Vérifions :

```
> evalm(subs(n=1,evalm(Bn)) - B),evalm(subs(n=2,evalm(Bn)) - B^2);
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

Mais :

```
> evalm(subs(n=0,evalm(Bn)) - B^0);
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & -1 \\ 1 & 0 & -1 & 0 \\ 1 & 0 & 0 & -1 \end{bmatrix}$$

Pourtant, les calculs effectués sont corrects, même pour $n = 0$ (X^0 est bien un polynôme ; on a bien de droit d'effectuer sa division euclidienne par le polynôme $\mu...$). La raison est que MAPLE a fait une simplification abusive dans le calcul de la liste des ordonnées : là où devrait figurer 0^n , il a écrit 0. C'est en général la même chose, sauf justement si $n = 0...$

4. Une telle matrice, si elle existe, est une racine carrée de la matrice B . Essayons donc la formule précédente avec $n = \frac{1}{2}$:

```
> C := subs(n=1/2,evalm(Bn));
```

$$C := \begin{bmatrix} I & 0 & 0 & 0 \\ -\frac{1}{2} + \frac{1}{2}I & 1 & 1 & -1 \\ I & 0 & 0 & 0 \\ I & 0 & 0 & 0 \end{bmatrix}$$

Vérifions :

```
> evalm(C^2-B);
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

Ça marche!

2 Décomposition $(P)A = LU$

2.1 $A = LU$

2.1.3 Application numérique

```
> LU:=proc(A)
  local n,i,j,c,L,U;
  n:=coldim(A);
  L:=diag(1$n);
  U:=A;
  for j from 1 to n-1 do
    for i from j+1 to n do
      c := U[i,j] / U[j,j];
      U := addrow(U,j,i,-c);
      L := addcol(L,i,j,c)
    od;
  od;
  evalm(L), evalm(U)
end;
```

Testons sur la matrice A :

```
> A:=matrix(3,3,[1,4,-2,2,6,1,3,5,3]);
(L,U) := LU(A);
```

$$L, U := \begin{bmatrix} 1 & 0 & 0 \\ 2 & 1 & 0 \\ 3 & \frac{7}{2} & 1 \end{bmatrix}, \begin{bmatrix} 1 & 4 & -2 \\ 0 & -2 & 5 \\ 0 & 0 & -\frac{17}{2} \end{bmatrix}$$

Vérifions :

```
> evalm(L&*U-A);
```

$$\begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

Pour la matrice B

```
> B:=matrix(4,4,[1,-1,0,5,-1,3,2,-4,2,4,5,17,0,-4,-5,7]);
LU(B);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ -1 & 1 & 1 & 0 \\ 2 & 3 & 1 & 0 \\ 0 & -2 & 1 & 1 \end{bmatrix} \begin{bmatrix} 1 & -1 & 0 & 5 \\ 0 & 2 & 2 & 1 \\ 0 & 0 & -1 & 4 \\ 0 & 0 & 0 & 5 \end{bmatrix}$$

L'appel de la fonction avec la matrice C provoque une erreur de division par 0 (le premier pivot est nul!) :

```
> LU(C);
```

Error, (in LU) numeric exception : division by zero

2.2 $PA = LU$

2.2.2 Pratique

La procédure `pgpivot` :

```
> pgpivot:=proc(A,j,n)
  local max, k,i,imax;
  max:=abs(A[j,j]);
  imax:=j;
  for i from j+1 to n do
    if abs(A[i,j]) > max then imax:=i; max:=abs(A[i,j]) fi
  od;
  imax;
end;
```

et la procédure PLU :

```
> PLU:=proc(A)
  local P,L,U,k,i,j,c,n;
  n:=coldim(A);
  P:=diag(1$n);
  L:=diag(1$n);
  U:=A;
  for j from 1 to n do
    k:=pgpivot(U,j,n);
    if k>j then
      P:=swaprow(P,j,k);
      L:=swaprow(swapcol(L,j,k),j,k);
      U:=swaprow(U,j,k)
    fi;
    for i from j+1 to n do
      c := U[i,j] / U[j,j];
      U := addrow(U,j,i,-c);
      L := addcol(L,i,j,c)
    od;
  od;
  evalm(P), evalm(L), evalm(U)
end;
```

2.2.3 Application numérique

Pour la matrice de l'énoncé, on trouve

```
> A:=matrix(3,3,[0,1,-1,1,3,1,2,6,1]):
  PLU(A);
```

$$\begin{bmatrix} 0 & 0 & 1 \\ 1 & 0 & 0 \\ 0 & 1 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ \frac{1}{2} & 0 & 1 \end{bmatrix}, \begin{bmatrix} 2 & 6 & 1 \\ 0 & 1 & -1 \\ 0 & 0 & \frac{1}{2} \end{bmatrix}$$